(define-constant ERR-UNAUTHORIZED u1)
(define-constant ERR-YOU-POOR u2)
(define-fungible-token whaley)
(define-data-var token-uri (optional (string-utf8 256)) none)
(define-constant contract-creator tx-sender)
(impl-trait 'SP1A14J4RY557V9BCBYQ0PQN3QMEVGAT17TYRCMV5.sip-010-trait-ft-standard.sip-010-trait)

;; SIP-010 Standard

(define-public (transfer (amount uint) (from principal) (to principal) (memo (optional (buff 34))))
    (begin
        (asserts! (is-eq from tx-sender)
            (err ERR-UNAUTHORIZED))

        (ft-transfer? whaley amount from to)
    )
)

(define-read-only (get-name)
    (ok "Whaley")
)

(define-read-only (get-symbol)
    (ok "WHALY")
)

(define-read-only (get-decimals)
    (ok u6)
)

(define-read-only (get-balance (user principal))
    (ok (ft-get-balance whaley user)
    )
)

(define-read-only (get-total-supply)
    (ok (ft-get-supply whaley)
    )
)

(define-public (set-token-uri (value (string-utf8 256)))
    (if 
        (is-eq tx-sender contract-creator) 
            (ok (var-set token-uri (some value))) 
        (err ERR-UNAUTHORIZED)
    )
)

(define-read-only (get-token-uri)
    (ok (var-get token-uri)
    )
)

;; send-many

(define-public (send-many (recipients (list 200 { to: principal, amount: uint, memo: (optional (buff 34)) })))
  (fold check-err
    (map send-token recipients)
    (ok true)
  )
)

(define-private (check-err (result (response bool uint)) (prior (response bool uint)))
  (match prior ok-value result
               err-value (err err-value)
  )
)

(define-private (send-token (recipient { to: principal, amount: uint, memo: (optional (buff 34)) }))
  (send-token-with-memo (get amount recipient) (get to recipient) (get memo recipient))
)

(define-private (send-token-with-memo (amount uint) (to principal) (memo (optional (buff 34))))
  (let
    ((transferOk (try! (transfer amount tx-sender to memo))))
    (ok transferOk)
  )
)

;; The Great $WHALY TGE / Total 10B

(begin
  (try! (ft-mint? whaley u10000000000000000 contract-creator)) 
)

;;
;;                                    ','. '. ; : ,','
;;                                      '..'.,',..'
;;                                         ';.'  ,'
;;                                          ;;
;;                                          ;'
;;                            :._   _.------------.___
;;                    __      :__:-'                  '--.
;;             __   ,' .'    .'             ______________'.
;;           /__ '.-  _\___.'          0  .' .'  .'  _.-_.'
;;              '._                     .-': .' _.' _.'_.'
;;                 '----'._____________.'_'._:_:_.-'--'